
Code currently sorts Bill of Materials on drawings by “Part number”.  Works well. (code shown below) 

 

However, team want to start to sort the Bill of Material so that Sub-Assemblies are Listed first and then parts.  

 

SolidWorks allows you to do it like this: 

 

This would be the desired result.  

 

I have not successfully been able to add that feature in the code. It would be great is code could be modified to create this result.  

I added this to the code, but it didn’t change the results.  

 

 

---- Here is the code---- 

 

Dim swApp As SldWorks.SldWorks 

Dim swModel As SldWorks.ModelDoc2 

Dim swDraw As SldWorks.DrawingDoc 



Dim swBomTable As SldWorks.BomTableAnnotation 

Dim swBomFeat As SldWorks.BomFeature 

Dim swSortData As SldWorks.BomTableSortData 

Dim swSelMgr As SldWorks.SelectionMgr 

Dim swFeat As SldWorks.Feature 

Dim vTables As Variant 

Dim boolstatus As Boolean 

Dim sortArray(2) As Integer 

Dim bWantGrp As Boolean 

 

Sub main() 

 

    ' Initialize SolidWorks application and active document 

    Set swApp = Application.SldWorks 

    Set swModel = swApp.ActiveDoc 

    Set swDraw = swModel 

    Set swSelMgr = swModel.SelectionManager 

    Set swFeat = swModel.FirstFeature 

     

    ' Toggle document property to set Top-level BOMS to 1 configuration 

    boolstatus = swModel.Extension.SetUserPreferenceToggle(swUserPreferenceToggle_e.swOneConfigOnlyTopLevelBom, 0, True) 

     

    ' Loop through features to find BOM feature 

    Do While Not swFeat Is Nothing 

        If swFeat.GetTypeName = "BomFeat" Then 

            swFeat.Select False 

            Set swBomFeat = swFeat.GetSpecificFeature2 

            vTables = swBomFeat.GetTableAnnotations 

            Set swBomTable = vTables(0) ' Get the first BOM table annotation 

            Exit Do ' Exit once the BOM is found 

        End If 

        Set swFeat = swFeat.GetNextFeature 

    Loop 

     

    ' Get BOM sorting data 

    Set swSortData = swBomTable.GetBomTableSortData 

     

    ' Specify the literal sort method 

    swSortData.SortMethod = swBomTableSortMethod_Literal 

    swSortData.ColumnIndex(0) = 1 ' Sort by the second column (index 1) 

    swSortData.Ascending(0) = True ' Sort in ascending order 

     

    ' Group rows by assemblies and parts 

    bWantGrp = True 

    If bWantGrp Then 

        sortArray(0) = swBomTableSortItemGroup_e.swBomTableSortItemGroup_Assemblies 

        sortArray(1) = swBomTableSortItemGroup_e.swBomTableSortItemGroup_Parts 

        sortArray(2) = swBomTableSortItemGroup_e.swBomTableSortItemGroup_None ' No further grouping 

        swSortData.ItemGroups = sortArray 

    End If 

     



    ' Allow re-numbering of items after sorting 

    swSortData.DoNotChangeItemNumber = False 

     

    ' Apply the sorting to the BOM 

    boolstatus = swBomTable.Sort(swSortData) 

     

    ' Clear selections 

    swModel.ClearSelection2 True 

 

End Sub 

 


